

THE ON-RAMP TO ROBOTICS WITH TI-INNOVATOR™ SYSTEM (TI-NSPIRE CX)

**Overview:** **Goals:**

In the first challenge, as an introduction to the color sensor, students use the color sensor on Rover to set the RGB LED on the Rover. In the second challenge students use the color sensor and a piece of colored construction paper to change Rover’s motion. In the final challenge, students revisit Olympus Mons from the first Unit, and search the surface of Mars for the presence of minerals by using the color sensor as the Rover drives over the colored minerals.

Students will:

1. use the color sensor mounted on the bottom of Rover to make measurements of color on the drive surface.
2. use the RGB LED mounted on the top of Rover.
3. use the color sensor to control the motion of the Rover.

**Background:**

What is Color?	Pure spectral colors of visible light are narrow wavelength bands of electromagnetic radiation that create a sensation when they strike the retina of the human eye. The color ranges indicated below are an approximation; the spectrum is continuous, with no clear boundaries between one color and the next. <b>violet</b> 380–450 nm, <b>blue</b> 450–475 nm, <b>cyan</b> 476–495 nm, <b>green</b> 495–570 nm, <b>yellow</b> 570–590 nm, <b>orange</b> 590–620 nm, <b>red</b> 620–750 nm
How do colors mix?	There are two types of color mixing, Additive and Subtractive. In both cases there are three primary colors, three secondary colors (colors made from 2 of the three primary colors in equal amounts), and one tertiary color made from all three primary colors.  1. <b>Additive</b> - Additive color mixing is the kind you get if you overlap colored spotlights in a dark room. The primary colors for additive color mixing are <b>red</b> , <b>green</b> and <b>blue</b> . This process is called additive because colored light is being added.  2. <b>Subtractive</b> - Subtractive color mixing is the kind you get with paints and pigments. Subtractive color mixing can be demonstrated with filters that absorb particular colors of light from white light. For example, white light possesses all of the spectral colors, the pigment in a red shirt will absorb or subtract out of green and blue light while reflecting the red light to the viewer’s eye. The primary colors for pigments are <b>cyan</b> , <b>magenta</b> and <b>yellow</b> . They are called primary because they absorb only one primary color from white light. For example yellow, a primary pigment, absorbs blue light and reflects red and green. The reflected red and green are perceived by the human eye as yellow.
Black and White	When all three primary color of light are added, white light is produced. Black is the absence of all light. Black pigment absorbs all incident light and reflects none to the eye, while white pigment absorbs none of the light and reflects all of the colors to the eye. This is also why black clothing is hot while white clothing is cool.
Rover’s RGB LED	The color RGB LED on the Rover has three parameters; the value of red, the value of green and the value of blue. For example Send “SET RV.COLOR 255 128 0” produces yellow illumination of the Rover’s LED. This RGB color space can be set to mimic most of the colors the eye can perceive.
Rover’s Color Sensor	The color sensor on the bottom of the rover detects pigment color. The white LED that accompanies the color sensor, shines a full spectrum white light onto the pigment. The color sensor then measures the values of the reflected red, green and blue light. For example, if Rover drives over a yellow colored piece of construction paper, the white LED will shine red, green and blue light upon the paper. The yellow pigment then subtracts out the blue light and reflects the red and green light. The sensor then measures the value of the red and green light. For example when the command “READ RV.COLORINPUT” when driving over yellow, the return values will be 255, 255, 0 indicating strong red and green light reflection and no blue light reflection.

**THE ON-RAMP TO ROBOTICS WITH TI-INNOVATOR™ SYSTEM (TI-NSPIRE CX)**
**Background Coding Commands**

Command	Example	Behavior
"READ RV.COLORINPUT.<color>"	Send "READ RV.COLORINPUT.RED"	Reads the color directly below the front of the Rover and returns that reading to the calculator when the program requests it with the Get command, in this example, just red is measured. The returned value ranges from 0 to 255 where 0 means there is none of the color component present, while 255 means the color is fully present. This command, in addition to red, also works for green, blue, and gray.
"READ RV.COLORINPUT"	Send "READ RV.COLORINPUT"	Reads the color directly below the front of the Rover and returns that reading to the calculator when the program requests it with the Get command. The returned values are 1 through 9 and are matched to particular colors. The color numbers are: 1-red, 2-green, 3-blue, 4-cyan, 5-magenta, 6-yellow, 7-black, 8-white, 9-gray.
Get variable	Get color	The Get command retrieves the value returned to the calculator from the Hub after a READ command is issued. The Get command must immediately follow the READ command. In the previous example the color measurement is returned and stored in the variable named color.
"SET RV.COLOR.RED <value>"	Send "SET RV.COLOR.RED 128"	Sets the color of Rover's RGB LED to a medium brightness red. The value can range from 0 to 255. This command also works for green and blue components of the LED.
getKey()	k=getKey()	Returns the last key pressed as a string to the variable named k. e.g. the number 1 key is returned as "1" and the escape key is returned as "esc" when the value of k is accessed.
"READ RV.WAYPOINT.CMDNUM"	Send "READ RV.WAYPOINT.CMDNUM"	The CMDNUM is used as a way to check if the Rover has finished driving its path. A value of 1 means the Rover has finished. A value of 0 means the Rover is still completing its path.

**Setup Rover:**

Students may work in groups of two or three. Choose an area that has at least 2 meters of clear uniform floor space. Carpeted flooring is less desirable than tile. If needed, drive mats may be use as a driving surface.

**Supplies:**

- 4" x 4" squares of red, green, blue, and black construction paper
- Roll of white paper, or white floor, or white drive mat
- Olympus Mons Drive Mat (optional)

**Student Activity**

**Challenge 1:** Continuously read the color sensor and use the color measurement to set the RGB LED on the Rover to match. Test your program by placing the Rover's color sensor on top of construction paper of various colors.

**Teacher Activity**
**Guidance during challenge 1:**

- The color sensor measures the value of the 3 color channels red, green and blue. The returned value will range from 0 to 255. The greater the value the more light of that particular color is reflected from the surface.
- For example, if rover were driving on a pure blue surface, the command "READ RV.COLORINPUT.BLUE" would return a value of 255, while on the same surface, the command

**THE ON-RAMP TO ROBOTICS WITH TI-INNOVATOR™ SYSTEM (TI-NSPIRE CX)**

"READ RV.COLORINPUT.RED" would return a value of 0, since there is no red light reflected from the surface. In this program the color sensor is continuously monitored using a While..EndWhile loop in conjunction with the getKey() command. Note: This control structure was used in Unit 2-Skill Builder 2-Challenge 1.

- Each cycle of the loop will read the three color channels, store the values from each into a unique variable and then use each variable to set the corresponding color channel of the RGB LED on the Rover.

**Sample Code:**

```
Define c1()=
Prgm
Send "CONNECT RV"
k:=""
While k≠"esc"
  Send "READ RV.COLORINPUT.RED"
  Get red
  Send "READ RV.COLORINPUT.GREEN"
  Get green
  Send "READ RV.COLORINPUT.BLUE"
  Get blue
  Send "SET RV.COLOR.RED eval(red)"
  Send "SET RV.COLOR.GREEN eval(green)"
  Send "SET RV.COLOR.BLUE eval(blue)"
  DispAt 3,"R= ",red,"G= ",green,"B= ",blue
  Wait 0.5
  k:=getKey()
EndWhile
EndPrgm
```

\*Please see the associated sample program, c1, in the file named "Unit 2 Skill Builder 3 Example Programs.tns" as a reference if help is needed.

**THE ON-RAMP TO ROBOTICS WITH TI-INNOVATOR™ SYSTEM (TI-NSPIRE CX)**

**Challenge 2:** Drive the Rover forward and have it stop when it drives over a piece of blue construction paper placed anywhere along the forward drive path.

**Guidance during challenge 2:**

- This activity makes use of the feature that Rover can do two things at one time. In this challenge, Rover will be driving forward for up to 5 meters. While driving forward, Rover will be continuously polling the color sensor and checking for the color blue.
- The TI BASIC program will stay in the While-EndWhile loop as long as the color is not equal to blue (number 3). As soon the Rover passes over the blue construction paper the While expression becomes false and the program exits the loop. The command following the EndWhile is the “RV STOP”, which immediately halts the Rover.

**Sample Code:**

```
Define c2()  
Prgm  
Send "CONNECT RV"  
color:=0  
Send "RV FORWARD 5 M"  
While color#3  
  Send "READ RV.COLORINPUT"  
  Get color  
  DispAt 3,"Color Number= ",color  
EndWhile  
Send "RV STOP"  
EndPrgm
```

\*Please see the associated sample program, c2, in the file named “Unit 2 Skill Builder 3 Example Programs.tns” as a reference if help is needed.

**Challenge 3:** Olympus Mons Mineral Challenge. Use the Olympus Mons drive mat and plot a course around the volcano that drives over the squares of colored construction paper attached to the mat. When the Rover drives over the


**Guidance during challenge 3:**

- This challenge requires the Rover to do two tasks at the same time. The Rover will drive a path around the Olympus Mons volcano and at the same time the Rover will be reading the color sensor to identify minerals on the surface of the path. The path will be programmed using the “RV TO XY” command and the grid printed on the Mars drive mat. The COLORINPUT sensor will be read as part

**THE ON-RAMP TO ROBOTICS WITH TI-INNOVATOR™ SYSTEM (TI-NSPIRE CX)**

following colors, display an appropriate message on the calculator such as:

- If green (2), display “Olivine Found!”
- If blue (3), display “Azurite Found!”
- If yellow (6), display “Sulphur Found!”
- If white (8), display “Calcium Carbonate Found!”



of a While loop that will execute while the Rover is still driving its path.

- Minerals are represented by the construction paper squares which should be centered above the colored x’s toward the edge of the Olympus Mons drive mat. For example, notice the blue construction paper placed above one of the x’s on the map to the left.
- If printing the drive mat is not possible, set up a 10 cm grid on a piece of 3’ x 3’ (or larger) butcher paper. For example, a coordinate of (3.5, -0.5) would be located 35 cm in the X direction and 5 cm in the -Y direction (see image for reference)
- Because construction paper can vary in color (black may appear as gray or vice versa), you should use the following program to test your colors using the color sensor on the rover.
- To do this, run the following program and pick up and move the rover from each of the colors of construction paper you plan to use being sure to place the color sensor above the paper. The program will report the color rover sees. Before students do this challenge, make sure they are aware of the color for each of the construction paper squares. Recall that each color has an associated number (1-red, 2-green, 3-blue, 4-cyan, 5-magenta, 6-yellow, 7-black, 8-white, 9-gray.)

**Color Test Code:**

```

Define colortest()
Prgm
Send "CONNECT RV"
color:=0
key:=""
While key!="esc"
    Send "READ RV.COLORINPUT"
Get color
DispAt 3,"Color Number=",color
Wait 0.5
key:=getKey()
EndWhile
EndPrgm
    
```

- Reading the RV.WAYPOINT.COMDNUM is used to determine if the Rover has completed driving its path. The CMDNUM value will be 0 if the Rover is still driving its path and 1 if the Rover has completed the path. In the program the While loop that reads the COLORINPUT sensor checks to

## THE ON-RAMP TO ROBOTICS WITH TI-INNOVATOR™ SYSTEM (TI-NSPIRE CX)

determine if the Rover is still driving the path with the conditional statement *cmdnum=0*. The program will run the While loop when *cmdnum=0* is true. The program will exit the While loop when the Rover is finished with its path and the CMDNUM value is 1, which means that *cmdnum=0* is false.

- Note: Find “READ RV.WAYPOINT.CMDNUM” on the Rover Read RV Path menu found at Menu/8: Hub/7: Rover (RV)/4: Read RV Path/3: RV.WAYPOINT.CMDNUM
- The initial value of 0 for *cmdnum* ensures the loop will run at least once.
- The program uses If..Then..Elseif commands to match color values with minerals. If..Then..Elseif commands are useful when there are multiple levels of mutually exclusive categories to check. Checking stops as soon as one of the Elseif statements is true. This saves the program from doing unnecessary checks.

```
Define c3()=
Prgm
Send "CONNECT RV"
Send "RV TO XY 3.5 -0.5"
Send "RV TO XY 5.5 2.5"
Send "RV TO XY 5 6"
Send "RV TO XY 2 6.5"
Send "RV TO XY -1.5 5.5"
Send "RV TO XY -1.5 2"
Send "RV TO XY 0 0"
cmdnum:=0
While cmdnum=0
  Send "READ RV.WAYPOINT.CMDNUM"
  Get cmdnum
  Send "READ RV.COLORINPUT"
Get color
DispAt 2,"Color Number=",color
If color=2 Then
DispAt 4,"Olivine Found"
```

**THE ON-RAMP TO ROBOTICS WITH TI-INNOVATOR™ SYSTEM (TI-NSPIRE CX)**

```
ElseIf color=3 Then
DispAt 5,"Azurite Found"
ElseIf color=6 Then
DispAt 6,"Sulphur Found"
ElseIf color=8 Then
DispAt 7,"Calcium Carbonate Found"
EndIf
EndWhile
EndPrgm
```

\*Please see the associated sample program, c3, in the file named "Unit 2 Skill Builder 3 Example Programs.tns" as a reference if help is needed.